
3. Corrections required
This section describes the TTCN changes required to ‘iwd-EUTRA-B2012-03_D13wk08’ release.
Change 1   cr_IMS_RoutingInfo

	Template name
	cr_IMS_RoutingInfo

	Reason for change
	The template cr_IMS_RoutingInfo is used to receive protected or unprotected IMS messages. The security field will be either protected or unprotected and was incorrectly set to omit.



	Summary of change
	Expected security set to *

	TTCN module
	IMS_ASP_Templates.ttcn

	MCC160 Comment
	


Before:

	  template IMS_RoutingInfo_Type cr_IMS_RoutingInfo(template (present) InternetProtocol_Type p_Protocol := ?,

                                                   template IMS_ProtectedUnprotected_Type p_ProtectedUnprotected := omit) := 
  {

    Protocol := p_Protocol,

    Security := p_ProtectedUnprotected,

    UE_Address := *,       // @sic R5s120907 change 1 sic@

    NW_Address := *        // @sic R5s120907 change 1 sic@

  };




After:

	  template IMS_RoutingInfo_Type cr_IMS_RoutingInfo(template (present) InternetProtocol_Type p_Protocol := ?,

                                                   template IMS_ProtectedUnprotected_Type p_ProtectedUnprotected := *) := 
  {

    Protocol := p_Protocol,

    Security := p_ProtectedUnprotected,

    UE_Address := *,       // @sic R5s120907 change 1 sic@

    NW_Address := *        // @sic R5s120907 change 1 sic@

  };




Change 2 
f_IMS_NotifyRequest_MessageHeader_A16

	 Function name
	f_IMS_NotifyRequest_MessageHeader_A16

	Reason for change
	v_Subscribe_From.addressField is of type NameAddr rather than SipUrl


	Summary of change
	Refer to NameAddr.SipUrl

	TTCN module
	IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before:

	function f_IMS_NotifyRequest_MessageHeader_A16(SUBSCRIBE_Request p_SubscribeRequest,

                                                 boolean p_CondA2,

                                                 integer p_CseqValue,

                                                 template (value) SubscriptionState p_SubscriptionState) runs on IMS_PTC return template (value) MessageHeader

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var template (value) MessageHeader v_MessageHeader_Notify := cs_MessageHeader_Dummy;

    var From v_Subscribe_From := p_SubscribeRequest.msgHeader.fromField;

    var SipUrl v_Subscribe_From_SipUrl := v_Subscribe_From.addressField.addrSpecUnion;
    var charstring v_Subscribe_From_ToTag := f_SIP_SemicolonParamList_GetParamValue(v_Subscribe_From.fromParams, c_tagId);

    var SipUrl v_Subscribe_To_SipUrl := p_SubscribeRequest.msgHeader.toField.addressField.addrSpecUnion;
    var charstring v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    var charstring v_Branch1 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Branch2 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Scscf := px_IMS_Scscf;

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps(not p_CondA2);   // @sic R5s120907 change 5 - Port_ps instead of Port_us sic@

    var template (value) ViaBody_List v_ViaBodyList;

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();

    var charstring v_ProtocolString := f_IMS_ConvertTransportProtocolToString(f_IMS_PTC_ImsInfo_GetTransportProtocol());

    v_ViaBodyList := {

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_NW_AddrStr, v_Port_ps), f_ViaParamsTX(v_Branch1, v_SigCompParam)),

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_Scscf),                 f_ViaParamsTX(v_Branch2))

    };

    v_MessageHeader_Notify.via         := cs_Via(v_ViaBodyList);

    v_MessageHeader_Notify.fromField   := cs_From(v_Subscribe_From_SipUrl, tsc_IMS_ToTagSubscribeDialog);  /* @sic R5-130750: same URI as in previous SUBSCRIBE's To header sic@ */

    v_MessageHeader_Notify.toField     := cs_ToWithTag(v_Subscribe_To_SipUrl, v_Subscribe_From_ToTag);     /* @sic R5-130750: same URI as in previous SUBSCRIBE's From header sic@

                                                                                                              NOTE: in fact the URIs in From and To are the same */

    v_MessageHeader_Notify.callId      := p_SubscribeRequest.msgHeader.callId;

    v_MessageHeader_Notify.cSeq        := cs_CseqDef(p_CseqValue, "NOTIFY");

    v_MessageHeader_Notify.contact     := cs_Contact(cs_SipUrl_HostPort(v_Scscf));     // @sic R5s120727 change 10 sic@

    v_MessageHeader_Notify.contentType := cs_ContentTypeDef ("application/reginfo+xml");

    v_MessageHeader_Notify.event       := cs_EventDef("reg");

    v_MessageHeader_Notify.maxForwards := cs_MaxForwardsDef(69);

    v_MessageHeader_Notify.subscriptionState := p_SubscriptionState;

    return v_MessageHeader_Notify;

  }


After:

	  function f_IMS_NotifyRequest_MessageHeader_A16(SUBSCRIBE_Request p_SubscribeRequest,

                                                 boolean p_CondA2,

                                                 integer p_CseqValue,

                                                 template (value) SubscriptionState p_SubscriptionState) runs on IMS_PTC return template (value) MessageHeader

  { /* Send NOTIFY acc. to 34.229-1 Annex A.1.6 */

    var template (value) MessageHeader v_MessageHeader_Notify := cs_MessageHeader_Dummy;

    var From v_Subscribe_From := p_SubscribeRequest.msgHeader.fromField;

    var SipUrl v_Subscribe_From_SipUrl := v_Subscribe_From.addressField.nameAddr.addrSpec;
    var charstring v_Subscribe_From_ToTag := f_SIP_SemicolonParamList_GetParamValue(v_Subscribe_From.fromParams, c_tagId);

    var SipUrl v_Subscribe_To_SipUrl := p_SubscribeRequest.msgHeader.toField.addressField.nameAddr.addrSpec;
    var charstring v_NW_AddrStr := f_IMS_PTC_NW_Address_GetHostFormat();

    var charstring v_Branch1 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Branch2 := f_IMS_PTC_ImsInfo_GetNextBranch();

    var charstring v_Scscf := px_IMS_Scscf;

    var integer v_Port_ps := f_IMS_PTC_GetPort_ps(not p_CondA2);   // @sic R5s120907 change 5 - Port_ps instead of Port_us sic@

    var template (value) ViaBody_List v_ViaBodyList;

    var template (omit) GenericParam v_SigCompParam := f_IMS_PTC_ImsInfo_GetSigCompParam();

    var charstring v_ProtocolString := f_IMS_ConvertTransportProtocolToString(f_IMS_PTC_ImsInfo_GetTransportProtocol());

    v_ViaBodyList := {

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_NW_AddrStr, v_Port_ps), f_ViaParamsTX(v_Branch1, v_SigCompParam)),

      cs_ViaBody(v_ProtocolString, cs_HostPort(v_Scscf),                 f_ViaParamsTX(v_Branch2))

    };

    v_MessageHeader_Notify.via         := cs_Via(v_ViaBodyList);

    v_MessageHeader_Notify.fromField   := cs_From(v_Subscribe_From_SipUrl, tsc_IMS_ToTagSubscribeDialog);  /* @sic R5-130750: same URI as in previous SUBSCRIBE's To header sic@ */

    v_MessageHeader_Notify.toField     := cs_ToWithTag(v_Subscribe_To_SipUrl, v_Subscribe_From_ToTag);     /* @sic R5-130750: same URI as in previous SUBSCRIBE's From header sic@

                                                                                                              NOTE: in fact the URIs in From and To are the same */

    v_MessageHeader_Notify.callId      := p_SubscribeRequest.msgHeader.callId;

    v_MessageHeader_Notify.cSeq        := cs_CseqDef(p_CseqValue, "NOTIFY");

    v_MessageHeader_Notify.contact     := cs_Contact(cs_SipUrl_HostPort(v_Scscf));     // @sic R5s120727 change 10 sic@

    v_MessageHeader_Notify.contentType := cs_ContentTypeDef ("application/reginfo+xml");

    v_MessageHeader_Notify.event       := cs_EventDef("reg");

    v_MessageHeader_Notify.maxForwards := cs_MaxForwardsDef(69);

    v_MessageHeader_Notify.subscriptionState := p_SubscriptionState;

    return v_MessageHeader_Notify;

  }




Change 3 
f_IMS_Authorization_CheckDigestResponse
	 Function name
	f_IMS_Authorization_CheckDigestResponse

	Reason for change
	Ordering of calls to bit2oct() and substr() is wrong.

	Summary of change
	Switched ordering. 

	TTCN module
	CommonIMS\IMS_CommonFunctions.ttcn

	MCC160 Comment
	


Before:

	  function f_IMS_Authorization_CheckDigestResponse(CommaParam_List p_DigestResponse,

                                                   bitstring p_XRES,

                                                   integer p_XRESLength) return boolean

  { // input is the comma separated parameter list: digestResponse from the Authorization header

    // and the value of XRES calculated in ts_AuthenticationInit

    // This function calculates the digest response (v_RequestDigest) for p_XRES according to

    // RFC 2617 and compares it with the response received from the UE (v_Dresponse)

    var octetstring v_METHOD    := char2oct("REGISTER");

    var octetstring v_Username  := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "username"));

    var octetstring v_Realm     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "realm"));

    var octetstring v_DigestUri := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "uri"));

    var octetstring v_Nonce     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nonce"));

    var octetstring v_NC        := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nc"));

    var octetstring v_CNonce    := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "cnonce"));

    var octetstring v_Qop       := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "qop"));

    var octetstring v_Passwd    := substr(bit2oct(p_XRES), 0, p_XRESLength); // @sic R5s120907 sic@
    var octetstring v_Dresponse := str2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "response"));

    var octetstring v_Secret;

    var octetstring v_HA2;

    var octetstring v_Data;

    var octetstring v_RequestDigest;

    var octetstring v_A1;

    var octetstring v_A2;

    var octetstring v_SecretData;

    // RFC 2617 3.2.2.2

    // A1 = unq(username-value) ":" unq(realm-value) ":" passwd

    v_A1 := v_Username & tsc_COLON_HEX & v_Realm & tsc_COLON_HEX & v_Passwd;

    // H(A1)

    v_Secret := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A1))));

    // RFC 2617 3.2.2.3

    // A2 = Method ":" digest-uri-value

    v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri;

    v_HA2 := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A2))));

    // RFC 2617 3.2.2.1

    //   unq(nonce-value)

    //   ":" nc-value

    //   ":" unq(cnonce-value)

    //   ":" unq(qop-value)

    //   ":" H(A2)

    v_Data := v_Nonce & tsc_COLON_HEX & v_NC & tsc_COLON_HEX & v_CNonce & tsc_COLON_HEX & v_Qop & tsc_COLON_HEX & v_HA2;

    //now apply KD function described in RFC 2617 section 3.2.1

    //KD(secret, data) = H(concat(secret, ":", data))

    v_SecretData := v_Secret & tsc_COLON_HEX & v_Data;

    v_RequestDigest := fx_MD5_Hex(v_SecretData);

    return (v_RequestDigest == v_Dresponse);

  }


After:

	function f_IMS_Authorization_CheckDigestResponse(CommaParam_List p_DigestResponse,

                                                   bitstring p_XRES,

                                                   integer p_XRESLength) return boolean

  { // input is the comma separated parameter list: digestResponse from the Authorization header

    // and the value of XRES calculated in ts_AuthenticationInit

    // This function calculates the digest response (v_RequestDigest) for p_XRES according to

    // RFC 2617 and compares it with the response received from the UE (v_Dresponse)

    var octetstring v_METHOD    := char2oct("REGISTER");

    var octetstring v_Username  := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "username"));

    var octetstring v_Realm     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "realm"));

    var octetstring v_DigestUri := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "uri"));

    var octetstring v_Nonce     := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nonce"));

    var octetstring v_NC        := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "nc"));

    var octetstring v_CNonce    := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "cnonce"));

    var octetstring v_Qop       := char2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "qop"));

    var octetstring v_Passwd    := bit2oct(substr(p_XRES, 0, p_XRESLength+1)); // @sic R5s120907 sic@ 
    var octetstring v_Dresponse := str2oct(f_SIP_CommaParamList_GetParamValue (p_DigestResponse, "response"));

    var octetstring v_Secret;

    var octetstring v_HA2;

    var octetstring v_Data;

    var octetstring v_RequestDigest;

    var octetstring v_A1;

    var octetstring v_A2;

    var octetstring v_SecretData;

    // RFC 2617 3.2.2.2

    // A1 = unq(username-value) ":" unq(realm-value) ":" passwd

    v_A1 := v_Username & tsc_COLON_HEX & v_Realm & tsc_COLON_HEX & v_Passwd;

    // H(A1)

    v_Secret := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A1))));

    // RFC 2617 3.2.2.3

    // A2 = Method ":" digest-uri-value

    v_A2 := v_METHOD & tsc_COLON_HEX & v_DigestUri;

    v_HA2 := char2oct(f_StringToLower(oct2str(fx_MD5_Hex(v_A2))));

    // RFC 2617 3.2.2.1

    //   unq(nonce-value)

    //   ":" nc-value

    //   ":" unq(cnonce-value)

    //   ":" unq(qop-value)

    //   ":" H(A2)

    v_Data := v_Nonce & tsc_COLON_HEX & v_NC & tsc_COLON_HEX & v_CNonce & tsc_COLON_HEX & v_Qop & tsc_COLON_HEX & v_HA2;

    //now apply KD function described in RFC 2617 section 3.2.1

    //KD(secret, data) = H(concat(secret, ":", data))

    v_SecretData := v_Secret & tsc_COLON_HEX & v_Data;

    v_RequestDigest := fx_MD5_Hex(v_SecretData);

    return (v_RequestDigest == v_Dresponse);

  }


Change 4 “f_IPCAN_StartProcedure”
	Function name
	f_IPCAN_StartProcedure

	Reason for change
	px_AuthN is one off.

	Summary of change
	Increase by one

	TTCN module
	34229_Model\IPCAN_Functions

	MCC160 Comment
	


Before:

	  function f_IPCAN_StartProcedure(IMS_TestProcedure_Type p_TestProcedure) runs on IPCAN_PTC return TriggerResult_Type
  {
    /* !!!! NOTE: there is no primitive at the 34.229 model to establish the RRC connection !!!! */
    var TriggerResult_Type v_Result := NORMAL;
    select (p_TestProcedure) {
      case (IPCAN_InitialRegistration) {
        f_IPCAN_IdleUpdated();
        if (not f_IPCANisEUTRA()) {
          f_IPCAN_ActivateDedicatedBearerContext(tsc_IMS_BearerContext1, tsc_IMS_BearerInfo1);     // activate (primary) PDP context
        }
        f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1], cms_IPCAN_IMS_RegInfo(cs_IPCAN_RegisteredInfo(px_AuthN))); // @sic Hellen sic@
      }

…


After:

	  function f_IPCAN_StartProcedure(IMS_TestProcedure_Type p_TestProcedure) runs on IPCAN_PTC return TriggerResult_Type
  {
    /* !!!! NOTE: there is no primitive at the 34.229 model to establish the RRC connection !!!! */
    var TriggerResult_Type v_Result := NORMAL;
    select (p_TestProcedure) {
      case (IPCAN_InitialRegistration) {
        f_IPCAN_IdleUpdated();
        if (not f_IPCANisEUTRA()) {
          f_IPCAN_ActivateDedicatedBearerContext(tsc_IMS_BearerContext1, tsc_IMS_BearerInfo1);     // activate (primary) PDP context
        }
        f_IMS_IPCAN_SendCoOrdMsg(IMS[tsc_Index_PDN1], cms_IPCAN_IMS_RegInfo(cs_IPCAN_RegisteredInfo(px_AuthN+1))); // @sic Hellen sic@
      }

…


Change 5 
f_IMS_SubscribeRequest_MessageHeaderRX
	 Function name
	f_IMS_SubscribeRequest_MessageHeaderRX

	Reason for change
	cr_ViaDef is strict in not expecting rport. As 24.229 is unclear in when rport is expected, UEs send rport in different scenarios.

	Summary of change
	Introduce a new template (see next change) relaxing restrictions on rport, and use the new template (this change) for Subscribe requests.

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration.ttcn

	MCC160 Comment
	


Before:

	  function f_IMS_SubscribeRequest_MessageHeaderRX(boolean p_IsGIBA := false,

                                                  EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  charstring p_ExpireTime := "600000") runs on IMS_PTC return template (present) MessageHeader

  { /* SUBSCRIBE acc. to 34.229-1 Annex A.1.4, A.1.5 or A.1.6

     * p_EventType:

     *   "reg"             - normal case acc. A.1.4

     *   "conference"      - "SUBSCRIBE for conference event package" acc. annex A.5.1; NOTE in this case p_SipUrlTo shall be present and set acc. to "px_FinalConferenceUri"

     *   "message-summary" - "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (p_IsGIBA) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, tsc_IMS_PortNumber_5060));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_Require;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/reginfo+xml") ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

  }


After:

	  function f_IMS_SubscribeRequest_MessageHeaderRX(boolean p_IsGIBA := false,

                                                  EventType p_EventType := "reg",

                                                  template SipUrl p_SipUrlTo := omit,

                                                  charstring p_ExpireTime := "600000") runs on IMS_PTC return template (present) MessageHeader

  { /* SUBSCRIBE acc. to 34.229-1 Annex A.1.4, A.1.5 or A.1.6

     * p_EventType:

     *   "reg"             - normal case acc. A.1.4

     *   "conference"      - "SUBSCRIBE for conference event package" acc. annex A.5.1; NOTE in this case p_SipUrlTo shall be present and set acc. to "px_FinalConferenceUri"

     *   "message-summary" - "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    var template (present) MessageHeader v_MessageHeader_Subscribe := cr_MessageHeader_Dummy;

    var IMS_ProtectedPorts_Type v_Protected;

    var template (present) PAccessNetworkInfo v_PAccessNetworkInfo := cr_PAccessNetworkInfoDef(f_IMS_PTC_RanType_GetString(), *);

    var template (present) SipUrl v_SipUrlFrom := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) SipUrl v_SipUrlTo;

    if (ispresent(p_SipUrlTo)) {

      v_SipUrlTo := p_SipUrlTo;         // NOTE: compiler warning cannot be avoided here

    } else {

      v_SipUrlTo := v_SipUrlFrom;

    }

    v_MessageHeader_Subscribe.cSeq        := cr_CseqDef(?, "SUBSCRIBE");

    v_MessageHeader_Subscribe.fromField   := cr_FromWithTag(v_SipUrlFrom);

    v_MessageHeader_Subscribe.toField     := cr_ToDef(v_SipUrlTo);

    v_MessageHeader_Subscribe.expires     := cr_Expires(p_ExpireTime);  // @sic R5-130548 sic@

    v_MessageHeader_Subscribe.callId      := cr_CallId(?);

    v_MessageHeader_Subscribe.maxForwards := cr_MaxForwardsDef;

    v_MessageHeader_Subscribe.event       := cr_EventDef(p_EventType);

    if (p_IsGIBA) {

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, *, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef(-, cr_HostPort(?));                                       // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, tsc_IMS_PortNumber_5060));         // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := omit;

      v_MessageHeader_Subscribe.proxyRequire := omit;

      v_MessageHeader_Subscribe.securityVerify := omit;

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo ifpresent;

    }

    else {

      v_Protected := f_IMS_PTC_Security_GetProtectedPorts();

      v_MessageHeader_Subscribe.route   := cr_RouteSubs (px_IMS_Pcscf, v_Protected.Port_ps, px_IMS_Scscf);

      v_MessageHeader_Subscribe.via     := cr_ViaDef2(-, cr_HostPort(?, v_Protected.Port_us));                   // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.contact := cr_Contact(cr_SipUrl_HostPort(?, v_Protected.Port_us));              // host address needs to be checked after receiving @sic R5s120727 change 6.1 sic@

      v_MessageHeader_Subscribe.require := cr_Require;

      v_MessageHeader_Subscribe.proxyRequire := cr_ProxyRequireDef;

      v_MessageHeader_Subscribe.securityVerify := f_IMS_PTC_BuildSecurityVerifyHeader();

      v_MessageHeader_Subscribe.pAccessNetworkInfo := v_PAccessNetworkInfo

    }

    select (p_EventType) {

      case ("reg") {                // normal case

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/reginfo+xml") ifpresent;

      }

      case ("conference") {         // "SUBSCRIBE for conference event package" acc. annex A.5.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/conference-info+xml");

        v_MessageHeader_Subscribe.expires := cr_Expires(?);

      }

      case ("message-summary") {    // "SUBSCRIBE for Message Waiting Indication package" acc. annex A.6.1

        v_MessageHeader_Subscribe.accept := cr_AcceptDef("application/simple-message-summary")

      }

    }

    return v_MessageHeader_Subscribe;

  }


Change 6 
cr_ViaDef
	 Template name
	cr_ViaDef

	Reason for change
	See previous change

	Summary of change
	Introduce new template cr_ViaDef2

	TTCN module
	CommonIMS\IMS_SIP_Templates

	MCC160 Comment
	


Before:

	  template Via cr_ViaDef(template (present) charstring p_Transport := ?,

                         template (present) HostPort p_HostPort := cr_HostPort(?, *),

                         template GenericParam p_Rport := omit) := //HR20130227 for TC 8.1

  { /* for initial request p_Rport may need to be set to "cr_GenericParam("rport", ?)" see 34.229-1 A.1.1 condition A1 */

    fieldName := VIA_E,

    viaBody := {

      cr_ViaBody(p_Transport, p_HostPort, f_ViaParamsRX(p_Rport))

    }

  };




After:

	template Via cr_ViaDef(template (present) charstring p_Transport := ?,

                         template (present) HostPort p_HostPort := cr_HostPort(?, *),

                         template GenericParam p_Rport := omit) := //HR20130227 for TC 8.1

  { /* for initial request p_Rport may need to be set to "cr_GenericParam("rport", ?)" see 34.229-1 A.1.1 condition A1 */

    fieldName := VIA_E,

    viaBody := {

      cr_ViaBody(p_Transport, p_HostPort, f_ViaParamsRX(p_Rport))

    }

  };

  template Via cr_ViaDef2(template (present) charstring p_Transport := ?,

                         template (present) HostPort p_HostPort := cr_HostPort(?, *)) :=
  { /* for initial request p_Rport may need to be set to "cr_GenericParam("rport", ?)" see 34.229-1 A.1.1 condition A1 */

    fieldName := VIA_E,

    viaBody := {

      cr_ViaBody(p_Transport, p_HostPort, {cr_GenericParam(c_branchId, pattern c_branchCookie & "*"),*})

    }

  };


Change 7 
f_IMS_SUBSCRIBE_Request
	 Function name
	f_IMS_SUBSCRIBE_Request

	Reason for change
	Usage of GRUU not fully implemented yet in TTCN

	Summary of change
	Do not enter f_IMS_PTC_CheckSipUrlForIPAddrOrFQDNOrGRUU by setting flag to false

	TTCN module
	CommonIMS\IMS_CommonProcedures_Registration

	MCC160 Comment
	


Before:

	  function f_IMS_SUBSCRIBE_Request(boolean p_IsGIBA := false,

                                   EventType p_EventType := "reg",

                                   charstring p_ExpireTime := "600000") runs on IMS_PTC return IMS_DATA_REQ

  { /* Receive SUBSCRIBE acc. to 34.229-1 Annex A.1.4 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var template (present) SipUrl v_SipUrlRequestUri := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) MessageHeader v_ExpectedMessageHeader;

    v_ExpectedMessageHeader := f_IMS_SubscribeRequest_MessageHeaderRX(p_IsGIBA, p_EventType, -, p_ExpireTime);

    IMS_Server.receive(car_IMS_Subscribe_Request(cr_SUBSCRIBE_Request(v_SipUrlRequestUri, v_ExpectedMessageHeader))) -> value v_IMS_DATA_REQ;

    v_MessageHeader := v_IMS_DATA_REQ.Request.Subscribe.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    f_IMS_MessageHeader_CheckTo(v_MessageHeader.toField);           // This cannot be checked in the template

    f_IMS_MessageHeader_CheckVia(v_MessageHeader, v_Protocol);      // @sic R5s120727 change 9 sic@

    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader, true);    // @sic R5s120727 change 9 sic@

    if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

      f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

    }

    return v_IMS_DATA_REQ;

  }


After:

	  function f_IMS_SUBSCRIBE_Request(boolean p_IsGIBA := false,

                                   EventType p_EventType := "reg",

                                   charstring p_ExpireTime := "600000") runs on IMS_PTC return IMS_DATA_REQ

  { /* Receive SUBSCRIBE acc. to 34.229-1 Annex A.1.4 */

    /* @sic R5-130750 change 4: specific IMPU for messages sent in DL sic@ */

    var IMS_DATA_REQ v_IMS_DATA_REQ;

    var MessageHeader v_MessageHeader;

    var InternetProtocol_Type v_Protocol;

    var template (present) SipUrl v_SipUrlRequestUri := f_IMS_PTC_ImsInfo_GetSipUriSubscriptionRX();

    var template (present) MessageHeader v_ExpectedMessageHeader;

    v_ExpectedMessageHeader := f_IMS_SubscribeRequest_MessageHeaderRX(p_IsGIBA, p_EventType, -, p_ExpireTime);

    IMS_Server.receive(car_IMS_Subscribe_Request(cr_SUBSCRIBE_Request(v_SipUrlRequestUri, v_ExpectedMessageHeader))) -> value v_IMS_DATA_REQ;

    v_MessageHeader := v_IMS_DATA_REQ.Request.Subscribe.msgHeader;

    v_Protocol := v_IMS_DATA_REQ.RoutingInfo.Protocol;

    f_IMS_MessageHeader_CheckTo(v_MessageHeader.toField);           // This cannot be checked in the template

    f_IMS_MessageHeader_CheckVia(v_MessageHeader, v_Protocol);      // @sic R5s120727 change 9 sic@

    f_IMS_MessageHeader_CheckContactAddr(v_MessageHeader, false);    // @sic R5s120727 change 9 sic@

    if (f_IMS_PTC_ImsInfo_SigCompStarted()) {

      f_IMS_MessageHeader_CheckSigComp(v_MessageHeader);

    }

    return v_IMS_DATA_REQ;

  }


Change 8 
fl_SIP_ParseURI
	Function name
	fl_SIP_ParseURI

	Reason for change
	According to TTCN3 standard, regexp function takes a template as second parameter

	Summary of change
	Fixed code

	TTCN module
	CommonIMS\IMS_SIP_Templates

	MCC160 Comment
	


Before:

	    function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* !!!! temporary implementation: type SipUrl needs to be replaced by proper definition which reflects generic uri definitions in a better way !!!! */

    var template (omit) UserInfo v_UserInfo := omit;

    var template (omit) charstring v_Host := omit;

    var charstring v_Scheme;

    var charstring v_Pattern;
    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    v_Pattern  := "([a-z]#(3,4)):(*)";

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern  := "(*)@(*)";
        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern  := "(*):(*)";
          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

        v_Host := v_HierPart;                                   // host is overloaded with the tel number

      }

      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_UserInfo := cs_UserInfo(v_HierPart);

      }

      case else {

        FatalError(__FILE__, __LINE__, "Invalid scheme");

      }

    }

    return cs_SipUrl(v_Scheme, v_UserInfo, v_Host);
  }


After:

	    function fl_SIP_ParseURI(charstring p_URI) return template (value) SipUrl

  { /* !!!! temporary implementation: type SipUrl needs to be replaced by proper definition which reflects generic uri definitions in a better way !!!! */

    var template (omit) UserInfo v_UserInfo := omit;

    var template (omit) charstring v_Host := omit;

    var charstring v_Scheme;

    var template charstring v_Pattern := pattern "([a-z]#(3,4)):(*)";
    var charstring v_HierPart;

    var charstring v_UserInfoString;

    var charstring v_UserName;

    v_Pattern  := "([a-z]#(3,4)):(*)";

    v_Scheme   := regexp(p_URI, v_Pattern, 0);

    v_HierPart := regexp(p_URI, v_Pattern, 1);

    select (v_Scheme) {

      case (c_sipScheme) {                                      // sip uri acc. to RFC 3261 cl. 19.1.1

        v_Pattern := pattern "(*)@(*)";
        v_UserInfoString := regexp(v_HierPart, v_Pattern, 0);

        if (v_UserInfoString == "") {                           // sip uri does not have user info e.g. "sip:biloxi.com" or sip:[2001:db8:1::139]

          v_Host := v_HierPart;

        } else {                                                // sip uri has user info

          v_Host := regexp(v_HierPart, v_Pattern, 1);

          v_Pattern := pattern "(*):(*)";
          v_UserName := regexp(v_UserInfoString, v_Pattern, 0);

          if (v_UserName == "") {                               // user info but no password e.g. "sip:alice@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserInfoString);

          } else {                                              // user info with password e.g. "sip:alice:secretword@atlanta.com"

            v_UserInfo := cs_UserInfo(v_UserName, regexp(v_UserInfoString, v_Pattern, 1));

          }

        }

      }

      case (c_telScheme) {                                      // tel uri acc. to RFC 3966

        v_Host := v_HierPart;                                   // host is overloaded with the tel number

      }

      case (c_urnScheme) {                                      // urn uri for emergency calls e.g. "urn:service:sos"

        v_UserInfo := cs_UserInfo(v_HierPart);

      }

      case else {

        FatalError(__FILE__, __LINE__, "Invalid scheme");

      }

    }

    return cs_SipUrl(v_Scheme, v_UserInfo, v_Host);
  }



Change 9 
f_MTC_IMS_MainLoop
	 Function name
	f_MTC_IMS_MainLoop

	Reason for change
	If only one PTC IMS started, then the condition is directly true and test case finishes at once

	Summary of change
	fixed

	TTCN module
	34229_Model\MTC_IMS_34229

	MCC160 Comment
	


Before:

	  function f_MTC_IMS_MainLoop(timer p_GuardTimer) runs on MTC_IMS

  { /* MTC main loop; called from the testcases

     * NOTE: it is assumed that the IMS_PTC is started just before calling this loop */

    var boolean v_IMS1_Done := (ptc_IMS1 != null);

    var boolean v_IMS2_Done := (ptc_IMS2 != null);

    while (not (v_IMS1_Done or v_IMS2_Done)) {
      alt {

        [not v_IMS1_Done] ptc_IMS1.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS1_Done := true;

        }

        [not v_IMS2_Done] ptc_IMS2.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS2_Done := true;

        }

        [] a_MTC_IMS_UpperTester(ims1,   IMS_MMI[tsc_Index_PDN1],   MMI_SYS);

        [] a_MTC_IMS_UpperTester(ims2,   IMS_MMI[tsc_Index_PDN2],   MMI_SYS);

        [] a_MTC_IMS_UpperTester(ipcan, IPCAN_MMI, MMI_SYS);

        [] p_GuardTimer.timeout {

          f_MTC_IMS_Stop(inconc);

        }

        [] any component.killed {

          f_MTC_IMS_Stop(inconc);

        }

      }

    }

    f_MTC_IMS_Stop(pass);   /* Note: "pass" overrules "none" only */

  }


After:

	  function f_MTC_IMS_MainLoop(timer p_GuardTimer) runs on MTC_IMS

  { /* MTC main loop; called from the testcases

     * NOTE: it is assumed that the IMS_PTC is started just before calling this loop */

    var boolean v_IMS1_Done := (ptc_IMS1 == null);

    var boolean v_IMS2_Done := (ptc_IMS2 == null);

    while (not (v_IMS1_Done and v_IMS2_Done)) {

      alt {

        [not v_IMS1_Done] ptc_IMS1.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS1_Done := true;

        }

        [not v_IMS2_Done] ptc_IMS2.done {  // terminate the loop when the IMS PTC(s) have done its work

          v_IMS2_Done := true;

        }

        [] a_MTC_IMS_UpperTester(ims1,   IMS_MMI[tsc_Index_PDN1],   MMI_SYS);

        [] a_MTC_IMS_UpperTester(ims2,   IMS_MMI[tsc_Index_PDN2],   MMI_SYS);

        [] a_MTC_IMS_UpperTester(ipcan, IPCAN_MMI, MMI_SYS);

        [] p_GuardTimer.timeout {

          f_MTC_IMS_Stop(inconc);

        }

        [] any component.killed {

          f_MTC_IMS_Stop(inconc);

        }

      }

    }

    f_MTC_IMS_Stop(pass);   /* Note: "pass" overrules "none" only */

  }

